(% i2)

```
/* Cantilever beam with a concentrated force at the free end */
/* Differential equations */
eq1: E * I * 'diff(ϕ(z), z, 2) - E / 2 * A * ('diff(a(z), z, 1) + ϕ(z)) =~0;
eq2: E / 2 * A * ('diff(ϕ(z), z, 1) + 'diff(a(z), z, 2)) = 0;
```

$$E\,I\left(\frac{d^2}{d\,z^2}\phi(z)\right)-\frac{A\,E\left(\frac{d}{d\,z}a(z)+\phi(z)\right)}{2}=0$$

$$\frac{A\,E\left(\frac{d}{d\,z}\phi(z)+\frac{d^2}{d\,z^2}a(z)\right)}{2}=0$$

(% i3)

```
/* Solving the system */
sol: desolve([eq1, eq2], [ϕ(z), a(z)]);
```

$$[\phi(z)=z\left(\left.\frac{d}{d\,z}\phi(z)\right|_{z=0}\right)+\frac{A\,z^2\left(\left.\frac{d}{d\,z}a(z)\right|_{z=0}\right)}{4\,I}+\frac{\phi(0)\,A\,z^2}{4\,I}+\phi(0),a(z)=-\left(\frac{z^2\left(\left.\frac{d}{d\,z}\phi(z)\right|_{z=0}\right)}{2}\right)-\frac{A\,z^3\left(\left.\frac{d}{d\,z}a(z)\right|_{z=0}\right)}{12\,I}+z\left(\frac{d}{d\,z}\right.$$

(% i4)

```
phi_sol: rhs(sol[1])$;
```

(% i5)

```
a_sol: rhs(sol[2])$;
```

(% i7)

```
/* Applying first two boundary conditions */
a_sol: subst([a(0) = 0, ϕ(0) = 0, 'diff(a(z), z, 1) = C, 'diff(ϕ(z), z, 1) = K], a_sol);
phi_sol: subst([a(0) = 0, ϕ(0) = 0, 'diff(a(z), z, 1) = C, 'diff(ϕ(z), z, 1) = K], phi_sol);
```

$$-\left(\frac{A\,C\,z^3}{12\,I}\right)-\frac{K\,z^2}{2}+C\,z$$

$$\frac{A\,C\,z^2}{4\,I}+K\,z$$

(% i8)

```
/* Third boundary condition */
bc1: ev(diff(phi_sol, z, 1), z = L) = 0;
```

$$\frac{A\,C\,L}{2\,I}+K=0$$

(% i9)

```
/* Fourth boundary condition */
bc2: A * E / 2 * (ev(phi_sol, z = L) + ev(diff(a_sol, z, 1), z = L)) = F;
```

$$\frac{A\,C\,E}{2}=F$$

(% i10)

```
/* Solving for C and K constants */
solve([bc1, bc2], [C, K])[1];
```

$$\left[C=\frac{2\,F}{A\,E},K=-\left(\frac{F\,L}{E\,I}\right)\right]$$

(% i11)

```
C_sol: rhs(solve([bc1, bc2], [C, K])[1][1])$
```

(% i12)

```
K_sol: rhs(solve([bc1, bc2], [C, K])[1][2])$
```

(% i13)

```
/* Final a(z) */
a_sol: factor(subst([C = C_sol, K = K_sol], a_sol));
```

$$-\left(\frac{F\,z\left(A\,z^2-3\,A\,L\,z-12\,I\right)}{6\,A\,E\,I}\right)$$

(% i14)

```
/* Final ϕ(z) */
phi_sol: factor(subst([C = C_sol, K = K_sol], phi_sol));
```

$$\frac{F\,z\left(z-2\,L\right)}{2\,E\,I}$$

(% i19)

```
/*****************************************************/
/* Cantilever beam with a distributed load */
/* Differential equations */
eq1: E * I * 'diff(ϕ(z), z, 2) - E / 2 * A * ('diff(a(z), z, 1) + ϕ(z)) = 0;
eq2: E / 2 * A * ('diff(ϕ(z), z, 1) + 'diff(a(z), z, 2)) = - p;
```

```
/* Solving the system */
sol: desolve([eq1, eq2], [ϕ(z), a(z)]);

phi_sol: rhs(sol[1])$;
a_sol: rhs(sol[2])$;
```

$$E\,I\left(\frac{d^2}{d\,z^2}\phi(z)\right)-\frac{A\,E\left(\frac{d}{d\,z}a(z)+\phi(z)\right)}{2}=0$$

$$\frac{A\,E\left(\frac{d}{d\,z}\phi(z)+\frac{d^2}{d\,z^2}a(z)\right)}{2}=-p$$

$$[\phi(z)=z\left(\left.\frac{d}{d\,z}\phi(z)\right|_{z=0}\right)+\frac{z^2\left(A\left(\left.\frac{d}{d\,z}a(z)\right|_{z=0}\right)+\phi(0)\,A\right)}{12\,I}+\frac{A\,z^2\left(\left.\frac{d}{d\,z}a(z)\right|_{z=0}\right)}{6\,I}-\frac{p\,z^3}{6\,E\,I}+\frac{\phi(0)\,A\,z^2}{6\,I}+\phi(0),a(z)=-\left(\frac{z^2\left(2\,A\right.}{}\right.$$

(% i21)

```
/* Applying first two boundary conditions */
a_sol: subst([a(0) = 0, ϕ(0) = 0, 'diff(a(z), z, 1) = C, 'diff(ϕ(z), z, 1) = K], a_sol);
phi_sol: subst([a(0) = 0, ϕ(0) = 0, 'diff(a(z), z, 1) = C, 'diff(ϕ(z), z, 1) = K], phi_sol);
```

$$\frac{p\,z^4}{24\,E\,I}-\frac{A\,C\,z^3}{12\,I}-\frac{(4\,p+2\,A\,E\,K)\,z^2}{24\,A\,E}-\frac{(2\,p+A\,E\,K)\,z^2}{6\,A\,E}-\frac{p\,z^2}{2\,A\,E}-\frac{K\,z^2}{4}+C\,z$$

$$-\left(\frac{p\,z^3}{6\,E\,I}\right)+\frac{A\,C\,z^2}{4\,I}+K\,z$$

–¿

(% i22)

```
/* Third boundary condition */
bc1: ev(diff(phi_sol, z, 1), z = L) = 0;
```

$$-\left(\frac{L^2\,p}{2\,E\,I}\right)+\frac{A\,C\,L}{2\,I}+K=0$$

(% i23)

```
/* Fourth boundary condition */
bc2: A * E / 2 * (ev(phi_sol, z = L) + ev(diff(a_sol, z, 1), z = L)) = 0;
```

$$\frac{A\,E\left(-\left(\frac{L\left(4\,p+2\,A\,E\,K\right)}{12\,A\,E}\right)-\frac{L\left(2\,p+A\,E\,K\right)}{3\,A\,E}-\frac{L\,p}{A\,E}+\frac{K\,L}{2}+C\right)}{2}=0$$

(% i24)

```
/* Solving for C and K constants */
solve([bc1, bc2], [C, K])[1];
```

$$\left[C=\frac{2\,L\,p}{A\,E},K=-\left(\frac{L^2\,p}{2\,E\,I}\right)\right]$$

(% i26)

```
C_sol: rhs(solve([bc1, bc2], [C, K])[1][1])$
K_sol: rhs(solve([bc1, bc2], [C, K])[1][2])$
```

(% i27)

```
/* Final a(z) */
a_sol: factor(subst([C = C_sol, K = K_sol], a_sol));
```

$$\frac{p\,z\left(A\,z^3-4\,A\,L\,z^2+6\,A\,L^2\,z-24\,I\,z+48\,I\,L\right)}{24\,A\,E\,I}$$

(% i28)

```
/* Final ϕ(z) */
phi_sol: factor(subst([C = C_sol, K = K_sol], phi_sol));
```

$$-\left(\frac{p\,z\left(z^2-3\,L\,z+3\,L^2\right)}{6\,E\,I}\right)$$

(% i30)

```
/* Simply supported beam with a distributed load */
/* Differential equations */
eq1: E * I * 'diff(ϕ(z), z, 2) - E / 2 * A * ('diff(a(z), z, 1) + ϕ(z)) = 0$;
eq2: E / 2 * A * ('diff(ϕ(z), z, 1) + 'diff(a(z), z, 2)) = - p$;
```

(% i31)

```
/* Solving the system */
sol: desolve([eq1, eq2], [ϕ(z), a(z)]);
```

$$\left[\phi\left(z\right)=z\left(\left.\frac{d}{d\,z}\,\phi\left(z\right)\right|_{z=0}\right)+\frac{z^2\left(A\left(\left.\frac{d}{d\,z}\,a\left(z\right)\right|_{z=0}\right)+\phi\left(0\right)\,A\right)}{12\,I}+\frac{A\,z^2\left(\left.\frac{d}{d\,z}\,a\left(z\right)\right|_{z=0}\right)}{6\,I}-\frac{p\,z^3}{6\,E\,I}+\frac{\phi\left(0\right)\,A\,z^2}{6\,I}+\phi\left(0\right),a\left(z\right)=-\left(\frac{z^2\left(2\,A\right.}{}\right.\right.$$

(% i33)

```
phi_sol: rhs(sol[1])$;
a_sol: rhs(sol[2])$;
```

(% i35)

```
/* Applying the first boundary condition */
a_sol: subst([a(0) = 0, ϕ(0) = J, 'diff(a(z), z, 1) = C, 'diff(ϕ(z), z, 1) = K], a_sol);
phi_sol: subst([a(0) = 0, ϕ(0) = J, 'diff(a(z), z, 1) = C, 'diff(ϕ(z), z, 1) = K], phi_sol);
```

$$\frac{p\,z^4}{24\,E\,I}-\frac{\left(A\,J+A\,C\right)z^3}{48\,I}-\frac{A\,J\,z^3}{16\,I}-\frac{A\,C\,z^3}{16\,I}-\frac{\left(4\,p+2\,A\,E\,K\right)z^2}{24\,A\,E}-\frac{\left(2\,p+A\,E\,K\right)z^2}{6\,A\,E}-\frac{p\,z^2}{2\,A\,E}-\frac{K\,z^2}{4}+C\,z$$

$$-\left(\frac{p\,z^3}{6\,E\,I}\right)+\frac{\left(A\,J+A\,C\right)z^2}{12\,I}+\frac{A\,J\,z^2}{6\,I}+\frac{A\,C\,z^2}{6\,I}+K\,z+J$$

(% i36)

```
/* Applying the second boundary condition */
bc1: subst(z = 0, diff(phi_sol, z, 1)) = 0;
```

$$K=0$$

(% i37)

```
/* Applying the third boundary condition */
bc2: ev(a_sol, z = L) = 0;
```

$$-\left(\frac{L^2\left(4\,p+2\,A\,E\,K\right)}{24\,A\,E}\right)-\frac{L^2\left(2\,p+A\,E\,K\right)}{6\,A\,E}+\frac{L^4\,p}{24\,E\,I}-\frac{L^2\,p}{2\,A\,E}-\frac{\left(A\,J+A\,C\right)L^3}{48\,I}-\frac{A\,J\,L^3}{16\,I}-\frac{A\,C\,L^3}{16\,I}-\frac{K\,L^2}{4}+C\,L=0$$

(% i38)

```
/* Applying the fourth boundary condition */
bc3: ev(diff(phi_sol, z, 1), z = L) = 0;
```

$$-\left(\frac{L^2\,p}{2\,E\,I}\right)+\frac{\left(A\,J+A\,C\right)L}{6\,I}+\frac{A\,J\,L}{3\,I}+\frac{A\,C\,L}{3\,I}+K=0$$

(% i39)

```
/* Solving for J, C and K constants */
solve([bc1, bc2, bc3], [J, C, K])[1];
```

$$\left[J=-\left(\frac{L^3\,p}{24\,E\,I}\right),C=\frac{\left(A\,L^3+24\,I\,L\right)p}{24\,A\,E\,I},K=0\right)$$

(% i42)

```
J_sol: rhs(solve([bc1, bc2, bc3], [J, C, K])[1][1])$
C_sol: rhs(solve([bc1, bc2, bc3], [J, C, K])[1][2])$
K_sol: rhs(solve([bc1, bc2, bc3], [J, C, K])[1][3])$
```

(% i43)

```
/* Final a(z) */
a_sol: factor(subst([J = J_sol, C = C_sol, K = K_sol], a_sol));
```

$$\frac{p\,z\,(z-L)\,(A\,z^2-A\,L\,z-A\,L^2-24\,I)}{24\,A\,E\,I}$$

(% i44)

```
/* Final φ(z) */
phi_sol: factor(subst([J = J_sol, C = C_sol, K = K_sol], phi_sol));
```

$$-\left(\frac{p\,(2\,z-L)\,(2\,z^2-2\,L\,z-L^2)}{24\,E\,I}\right)$$